
TITLE: 
Indentation problem. 
PROBLEM: 
This problem was suggested by EDF as a check of the code capability to model a classical contact 
problem by the contact models available, in particular by the pinballs method. The ideal problem is a 
static one. A rigid spherical indenter is forced to penetrate into an elastic perfectly plastic half space. 
The computed result is the resultant of the contact forces in the indentation direction as a function of 
time. 
An approximated analytical solution exists, due to Johnson. 
 MESH: 
The 2D model is axisymmetric and uses 1694 elements Q42L for the piece (half space represented by a 
square region of side equal to 3 times the indenter radius), and one PMAT element for the indenter. 
MATERIALS: 
The VM23 material is assigned to the piece, with zero plastic hardening. The MASS element is 
assigned to the indenter. 
BOUNDARY CONDITIONS: 
All nodes on the axis of symmetry are blocked in the radial direction. The base of the piece is blocked 
in all directions. Contact is represented by the pinball model. A single pinball with radius equivalent to 
that of the indenter is associated with the PMAT element. The region of the piece likely to come in 
contact is filled by parent pinballs with a hierarchy level of 4 to get accurate contact resolution. 
LOADING: 
The indenter is pushed into the piece at constant imposed speed (linear displacement in time), until it 
reaches an indentation depth equal to the radius of the indenter. 
INITIAL CONDITIONS 
The indenter has an initial velocity equal to the imposed indentation velocity (to avoid an initial error in 
the energy balance). 
CALCULATION: 
The calculation is performed up to 50 ms. At the final time, the indenter has reached a depth equal to 
its radius. 
RESULTS: 
The computed resultant contact force is in good agreement with the approximate analytical solution up 
to an indentation depth equal to approximately 1/2 of the indenter radius. For larger indentations, the 
analytical (linear) solution is no longer valid. 

POST-TREATMENT 
Several animations of the computed results from this calculation are available on the EUROPLEXUS 
Consortium Web site. 
 
REFERENCES: 
The indentation problem is detailed in the following EDF document: 
N. Tardieu, B. Serre: "Indentation elasto-plastique d'un bloc par un indenteur spherique elastique", 
Code_Aster, Manuel de Validation, Report V6.04.506, December 2002. 
This calculation is detailed in: 
F. Casadei: "Validation of the EUROPLEXUS Pinball Impact-Contact Model on an Indentation 
Problem", Technical Note in press, July 2007. (available on the EUROPLEXUS Consortium Web site). 
 
Numerical Solutions 
 
INDE10 
2D axisymmetric solution. The mesh generation file is: 
 
*%siz 50 
opti echo 1; 
opti dime 2 elem qua4; 
p0 = 0 0; 
p1 = 0 -1.5; 
p2 = 1.5 -1.5; 
p3 = 1.5 0; 
pc = 0 0.5; 
p13 = 0.375 0; 
tol = 1.e-5; 
piece = dall (p0 d 40 p1) (p1 d 40 p2) (p2 d 40 p3) (p3 d 40 p0) plan; 
sphere = manu poi1 pc; 
opti trac psc ftra 'vl_jrc_inde10_msh.ps'; 
trac qual (piece et sphere); 

tout = piece et sphere; 
axe = tout poin droi p1 pc tol; 
base = piece poin droi p1 p2 tol; 
cp = p0 d 10 p13; 
elim tol (piece et cp); 
c_p = piece elem appu larg cp; 
mesh = tout et axe et base et c_p; 
tass mesh; 
opti sauv form 'vl_jrc_inde10.msh'; 
sauv form mesh; 
list (nbno mesh); 
list (nbel mesh); 
fin; 

 

The input file is: 
 
INDE - 10 
ECHO 
!CONV WIN 
NONL AXIS 
CAST mesh 
DIME 
  PT2L 1682 
  Q42L 1694 
  PMAT 1 
  ZONE 2 
TERM 
GEOM  
  Q42L piece PMAT sphere 
TERM 
COMP EPAI 1.0 LECT piece sphere TERM 
     COUL rose LECT piece TERM 
          turq LECT sphere TERM 
MATE MASS 650.D0 
          LECT sphere TERM 

     VM23 RO 7800 YOUN 2.1E11 NU 0.3 ELAS 5.E7 
          TRAC 2 5.E7 2.38095E-4 5.E7 1.0 
          LECT piece TERM 
OPTI PINS CNOR NCOL RCEL 
LINK COUP 
     BLOQ 1 LECT axe base TERM 
          2 LECT base TERM 
     DEPL 2 -1.D0 FONC 1 LECT sphere TERM 
     PINB BODY DIAM 1.D0 LECT sphere TERM 
          BODY MLEV 4    LECT c_p TERM           
FONC 1 TABL 2 0. 0. 2. 20. 
INIT VITE 2 -10. LECT sphere TERM 
REGI 'r_piece' RESU LECT c_p  TERM 
     'r_base'  RESU POIN LECT base TERM 
ECRI DEPL VITE FEXT TFRE 0.005 
     FICH ALIC TFRE 25.E-5 
OPTI CSTA 0.5 LOG 1 
CALCUL TINI 0. TFIN 50.E-3 
FIN 

 
 
 
 
 

The initial configuration (with parent pinballs shown) and the final configuration 
(with contacts shown) are: 

The final velocities and displacement norm: 
 

 
The reaction force is: 

 
INDE13 
3D solution using only hexahedra. The model is not perfectly axisymmetric. The 
mesh generation file is: 
 
 
*%siz 50 
opti echo 1; 
opti dime 3 elem cub8; 
p0 = 0 0 0; 
p1 = 0 0 -1.5; 
p2 = 1.5 0 -1.5; 
p3 = 1.5 0 0; 
pc = 0 0 0.5; 
p13 = 0.375 0 0; 
vy = 0 1.5 0; 
py = 0 0.375 0; 
tol = 1.e-5; 
piece2 = dall (p0 d 20 p1) (p1 d 20 p2) (p2 d 20 p3) (p3 d 20 p0) plan; 
piece = piece2 volu tran 20 vy; 
sphere = manu poi1 pc; 
opti trac psc ftra 'vl_jrc_inde13_msh.ps'; 
trac cach qual (piece et sphere); 
tout = piece et sphere; 
blox = tout poin plan p1 p0 (p1 plus vy) tol; 
bloy = tout poin plan p1 p0 p2 tol; 
base = piece poin plan p1 p2 (p1 plus vy) tol; 
p13y = p13 plus py; 
cp = dall (p0 d 5 p13) (p13 d 5 p13y) 
          (p13y d 5 py) (py d 5 p0) plan; 
elim tol (piece et cp); 
c_p = piece elem appu larg cp; 
mesh = tout et blox et bloy et base et c_p; 
tass mesh; 
opti sauv form 'vl_jrc_inde13.msh'; 
sauv form mesh; 
list (nbno mesh); 
list (nbel mesh); 
fin; 
 
The input file is: 
INDE - 13 

ECHO 
!CONV WIN 
NONL TRID 
CAST mesh 
DIME 
  PT3L 9262 
  CUB8 9361 
  PMAT 1 
  ZONE 2 
TERM 
GEOM  
  CUB8 piece PMAT sphere 
TERM 
COMP COUL rose LECT piece TERM 
          turq LECT sphere TERM 
MATE MASS 1021.02D0 
          LECT sphere TERM 
     VMIS PARF RO 7800 YOUN 2.1E11 NU 0.3 ELAS 5.E7 
          LECT piece TERM 
OPTI PINS CNOR NCOL RCEL 
LINK COUP 
     BLOQ 1 LECT blox base TERM 
          2 LECT bloy base TERM 
          3 LECT base TERM 
     DEPL 3 -1.D0 FONC 1 LECT sphere TERM 
     PINB BODY DIAM 1.D0 LECT sphere TERM 
          BODY MLEV 4    LECT c_p TERM           
FONC 1 TABL 2 0. 0. 2. 20. 
INIT VITE 3 -10. LECT sphere TERM 
REGI 'r_piece' RESU LECT c_p  TERM 
     'r_base'  RESU POIN LECT base TERM 
ECRI DEPL VITE FEXT TFRE 0.005 
     FICH ALIC TFRE 25.E-5 
OPTI CSTA 0.5 LOG 1 
CALCUL TINI 0. TFIN 50.E-3 
FIN 

 

 

Approximate 
analytical 
solution 

The initial configuration (with parent pinballs shown) and the final configuration) are: 
 

 
 

The final velocities and displacement norm: 
 

 
 
The final deformed shape and the reaction force are: 
 

 

Approximate 
analytical 
solution 



 5

TEST14 
 
3D solution using an axisymmetric mes (hexahedra and prisms). The model is 
perfectly axisymmetric. The mesh generation file is: 
 
*%siz 50 
opti echo 0; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxordpoi.proc'; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxrota3d.proc'; 
opti echo 1; 
opti dime 3 elem cub8; 
p0 = 0 0 0; 
p1 = 0 0 -1.5; 
p2 = 1.5 0 -1.5; 
p3 = 1.5 0 0; 
pc = 0 0 0.5; 
p13 = 0.375 0 0; 
vy = 0 1.5 0; 
py = 0 0.375 0; 
tol = 1.e-5; 
c1 = p1 d 20 p2; 
ax = p1 d 20 p0; 
base ier = pxrota3d c1 18 90.0 ax tol; 
oubl ax; 
piece = base volu tran 20 (0 0 1.5); 
p_cube = piece elem cub8; 
p_pris = piece elem pri6; 

sphere = manu poi1 pc; 
opti trac psc ftra 'test14_msh.ps'; 
trac cach qual (piece et sphere); 
tout = piece et sphere; 
blox = tout poin plan p1 p0 (p1 plus vy) tol; 
bloy = tout poin plan p1 p0 p2 tol; 
c2 = p0 d 5 p13; 
ax2 = p0 d 1 pc; 
cp ier = pxrota3d c2 18 90.0 ax2 tol; 
oubl ax2; 
elim tol (piece et cp); 
c_p = piece elem appu larg cp; 
mesh = tout et blox et bloy et base et c_p; 
tass mesh; 
opti sauv form 'test14.msh'; 
sauv form mesh; 
list (nbno mesh); 
list (nbel mesh); 
list (nbel p_pris); 
list (nbel p_cube); 
fin; 

 

The input file is: 
 
TEST - 14 
ECHO 
!CONV WIN 
NONL TRID 
CAST mesh 
DIME 
  PT3L 8002 
  PR6  360 
  CUB8 8192 
  PMAT 1 
  ZONE 3 
TERM 
GEOM  
  PR6 p_pris CUB8 p_cube PMAT sphere 
TERM 
COMP COUL rose LECT piece TERM 
          turq LECT sphere TERM 
MATE MASS 1021.02D0 
          LECT sphere TERM 

     VMIS PARF RO 7800 YOUN 2.1E11 NU 0.3 ELAS 5.E7 
          LECT piece TERM 
OPTI PINS CNOR NCOL RCEL 
LINK COUP 
     BLOQ 1 LECT blox base TERM 
          2 LECT bloy base TERM 
          3 LECT base TERM 
     DEPL 3 -1.D0 FONC 1 LECT sphere TERM 
     PINB BODY DIAM 1.D0 LECT sphere TERM 
          BODY MLEV 4    LECT c_p TERM           
FONC 1 TABL 2 0. 0. 2. 20. 
INIT VITE 3 -10. LECT sphere TERM 
REGI 'r_piece' RESU LECT c_p  TERM 
     'r_base'  RESU POIN LECT base TERM 
ECRI DEPL VITE FEXT TFRE 0.005 
     FICH ALIC TFRE 25.E-5 
OPTI CSTA 0.5 LOG 1 
CALCUL TINI 0. TFIN 50.E-3 
FIN

 
 
 
 
 
The initial configuration (with parent pinballs shown) and the final velocities are: 

The final displacement norm and deformed shape: 

 
 
The reaction force is: 

 
 
Comparison of all solutions: 
 

 

Approximate 
analytical 
solution 

Approximate 
analytical 
solution 


